

2025-10-23 - Microcomputers

Volatile

- Any variable can be declared volatile.
 - `volatile int c`
 - `volatile int *pc;`
- Tells C compiler never to trust recently-accessed value for the variable; always re-read memory to get latest value.

Ex.

`d = a + b * c + b / c;`

- Most compilers only read 'b' from memory once on any subsequent appearances for the variable 'b'
- When 'c' is declared volatile, C compiler must read the value of 'c' from memory every time it appears in the program; it is not allowed to re-use a recently obtained value.

SimpleCpu ISA

3 types of instructions

Arithmetic:

`ADD rA, rB`
`SUB rA, rB`
`MUL rA, rB`
`AND rA, rB`

Data Movement:

`MOV rA, rB // rA = rB`
`MOV rA, IMM // rA = IMM`

$\text{IMM} \in \{-1, 1\}$

Memory access and Output

`LOAD rA, [rB] // rA = Mem[rB]`
`STORE rA, [rB] // Mem[rB] = rA, destination of rB assigned to rA`

$0 - 127 \rightarrow \text{memory}$

$\geq 128 \rightarrow \text{output reg (hex display)}$

Hardware:

$[1:0] \text{ Data} \rightarrow \text{ALU (SW}[4:3] \text{ to control operations)} \rightarrow \text{LED}[5:0], \text{ outputs}$

`SW[4:3]:`

- `00` $\rightarrow +$
- `01` $\rightarrow -$
- `10` $\rightarrow *$
- `11` $\rightarrow \&$

`LED[5:0]:`

-

Register file:

SW[9]: wA (decoder / enabler)

- 0: write to memory, do not write to any Register
- 1: write to register, designated by rA

SW[8:7]: rA

SW[6:5]: rB

Mux: SW[1:0]: Display result

- 01: for arithmetic operations

ex. MUL r0, r3

- SW[9]: 1
- SW[8:7]: 00
- SW[6:5]: 11
- SW[4:3]: 10 (ALUOp)
- SW[1:0]: 01

Instruction Sequence

- Given a starting address and load signal
- Address sent to Instruction Memory


Instruction memory stores "instruction words" ie. bits that control the CPU circuit (controls SW[9:0])

- Next instruction is stored at "PC+1"
- User counter to generate PC, PC+1, PC+2

If/else:

- Branches
 - Bxx LOCATION
 - Bxx = branch on condition xx, one of $\{C, B, VA, VS, N, Z\}$
- Jumps
 - JMP LOCATION //always go to location

2025-10-23 - RISC-V ISA



- For input select = [1, 1, 1], move to next instruction. (Next state = [x,x,x,x])
- For state 1, if overflow = 1, next state -> 3. Else + 1 (next state -> 2)
- For state 2, we want it only to go to next state -> 4

Chapter 6 - RISC-V ISA

- To negate a , it is essentially $0 - a$, so most instructions only provide subtraction instructions, not negation.

Architecture: Programmer's view of computer (defined by instructions, operand locations).

Microarchitecture: How to implement an architecture in hardware.

Assembly Language: Human-readable (perhaps) format of computer instructions.

Machine Language: Computer-readable format (binary)

RISC-V Assembly:

Addition:

```
# s0 = a, s1 = b, s2 = c
add a, b, c # b and c are operated on, a is the result.
```

Subtraction:

```
sub a, b, c
```

Simplicity favors regularity

- Consistent instruction format.
- Same number of operands (two sources, one destination)
- Easier to encode and handle in hardware

Ex. $a = b + c - d$

In assembly:

```
add t, b, c # t = b + c
sub a, t, d # a = t - d
```

Operands

Operand location: Physical location on hardware ie Registers, memory, constants.

Registers are faster than memory.

“32-bit architecture”: Uses 32-bit registers, operates on 32-bit data.

| Name | Register Number | Usage |
|--------------|-----------------|------------------------------------|
| zero | x0 | Constant value 0 |
| ra | x1 | Return address |
| sp | x2 | Stack pointer |
| gp | x3 | Global pointer |
| tp | x4 | Thread pointer |
| t0-2 | x5-7 | Temporaries |
| s0/fp | x8 | Saved register / Frame pointer |
| s1 | x9 | Saved register |
| a0-1 | x10-11 | Function arguments / return values |
| a2-7 | x12-17 | Function arguments |
| s2-11 | x18-27 | Saved registers |
| t3-6 | x28-31 | Temporaries |

- x0 is always 0
- Jump instruction can only write to x1

Registers

- Can use name (ra, zero or x0, x1, etc.)
- Convention:
 - s0 - s11: holds variables
 - t0 - t6: hold temporary variables

Can use immediate instructions.

ex. addi


```
# s0 = a, s1 = b
addi s0, s1, 6
```

Memory

- Much slower than registers.
- Too much data to fit into 32 registers
- Commonly used variables kept in registers
- Cache: faster memory with smaller size in between memory and register speed

Word-addressable memory:

- Each 32-bit data word has a unique address



- To read from memory, call “load word” = lw

```
lw t1, 5(s0)
lw destination, offset(base)
```

- Add base address (s0) to the offset (5)
- address = (s0 + 5)
- t1 holds the data value at address (s0 + 5)
- *any register* may be used as base address

ex. Read word of data at memory address 1 into s3.

```
lw s3, 1(zero) # read memory word 1 into s3
```

- To write to memory, call “store” = sw

ex. Write value in t4 to memory address 3

```
sw t4, 0x3(zero) # write value in t4 to memory word 3
```

- Add base address (zero) to offset (0x3)
- Address: (0 + 0x3) = 3
- Result: Writes the

Byte-addressable Memory

- Each data byte has unique address

- Load/store words or single bytes: load byte lb and store byte sb
- 32-bit word = 4 bytes, so word address *increments by 4*

The diagram illustrates memory organization across four columns: **Byte Address**, **Word Address**, **Data**, and **Word Number**.

- Byte Address:** A 5x4 grid showing addresses from 13 down to 3. The top row (address 13) contains letters F, E, D, C. The bottom row (address 3) contains letters A, B, C, D. The leftmost column is labeled **MSB** and the rightmost column is labeled **LSB**.
- Word Address:** A single column listing word addresses: 000000010, 00000000C, 000000008, 000000004, and 000000000.
- Data:** A 5x4 grid showing data values for each word address.
 - Word 4 (address 10): C D 1 9, A 6 5 B
 - Word 3 (address C): 4 0 F 3, 0 7 8 8
 - Word 2 (address 8): 0 1 E E, 2 8 4 2
 - Word 1 (address 4): F 2 F 1, A C 0 7
 - Word 0 (address 0): A B C D, E F 7 8
- Word Number:** A single column labeling the rows as **Word 4**, **Word 3**, **Word 2**, **Word 1**, and **Word 0**.

A double-headed arrow at the bottom indicates the width of the data field, labeled **width = 4 bytes**.

- **Example:** store the value held in t7 into memory address $0x10$ (16)
 - if t7 holds the value $0xAABBCCDD$, then after the sw completes, word 4 (at address $0x10$) in memory will contain that value

RISC-V assembly code

```
sw t7, 0x10(zero)    # write t7 into address 16
```

The diagram illustrates memory organization across four dimensions:

- Byte Address:** A 4x4 grid of addresses from 13 down to 0. The top row contains 13, 12, 11, 10; the second row F, E, D, C; the third row B, A, 9, 8; and the bottom row 7, 6, 5, 4. Below the grid, 'MSB' is aligned under address 13 and 'LSB' under address 0.
- Word Address:** A column of five red binary values: 00000010, 0000000C, 00000008, 00000004, and 00000000.
- Data:** A 4x4 grid of hexadecimal pairs. Row 4: A A B B C C D D; Row 3: 4 0 F 3 0 7 8 8; Row 2: 0 1 E E 2 8 4 2; Row 1: F 2 F 1 A C 0 7; Row 0: A B C D E F 7 8.
- Word Number:** A column of labels: Word 4, Word 3, Word 2, Word 1, and Word 0.


A double-headed arrow at the bottom indicates a width of 4 bytes between the start of the first word (address 13) and the start of the fourth word (address 0).

- Alignment: word address is not multiple of 4

```
lw t1, 2(zero) // 0xAC_07_AB_CD
```

- That is Little-Endian, what about Big-Endian?

```
lw t1, 2(zero) // 0xEF_78_F2_F1
```



Constants

12-bit signed constants (immediates) using addi:

Assembly:

```
# s0 = a, s1 = b
addi s0, zero, -372
addi s1, s0, 6
```

C:

```
// int is a 32-bit signed word
int a = -372;
int b = a + 6;
```

However, for assembly this does not work for values over 12 bits.

For these cases, use lui and addi.

- lui puts an immediate in the upper 20 bits of destination and 0s in lower 12 bits

Assembly:

```
# s0 = a
lui s0, 0xFEDC8
addi s0, s0, 0x765
```

C:

```
int a = 0xFEDC8765;
```

- If **bit 11** of 32-bit constant is **1**, increment upper 20 bits by **1** in `lui`

C Code

```
int a = 0xFEDC8EAB;
```

Note: -341 = 0xEAB

RISC-V assembly code

```
# s0 = a
lui s0, 0xFEDC9      # s0 = 0xFEDC9000
addi s0, s0, -341    # s0 = 0xFEDC9000 + 0xFFFFFEAB
                      #      = 0xFEDC8EAB
```

Logic/Shift instructions

- `and`: useful for bit masking. ex.
 - `0xF234012F AND 0X000000FF = 0x0000002F`
 - masks the last 2 bytes
- `or`: useful for combining bit fields
 - Combine `0xF2340000` with `0x000012BC`:
 - `0xF2340000 OR 0X000012BC = 0xF23412BC`
- `xor`: useful for inverting bits:
 - `A XOR -1 = NOT A (-1 = 0xFFFFFFFF)`
- `sll`: shift left logical
 - `slli t0, t1, 23` $\implies t0 = t1 \ll 23$
- `srl`: shift right logical (Inserts zeroes)
 - `srli`: immediate
- `sra`: shift right arithmetic (Preserves sign bit)
 - `srai`: immediate

for non-immediate shifts, ie `sll t0, t1, t2` only takes the least 5 significant bits of `t2`.

Multiplication and Division

32 x 32 multiplication \rightarrow 64 bit result

```
mul s3, s1, s2 # s3 = lower 32 bits of result
mulh s4, s1, s2 # s4 = higher 32 bits of result
```

32-bit division \rightarrow 32 bit quotient and remainder

- `div s3, s1, s2`
- `rem s4, s1, s2`

Branching

Conditional branches:

- `beq`: branch if equal
- `bne`: branch if not equal

- blt: branch if less than
- bge: branch if greater than or equal

Unconditional:

- j: jump
- jr: jump register
- jal: jump and link
- jalr: jump and link register

RISC-V assembly

```
addi s0, zero, 4      # s0 = 4
addi s1, zero, 1      # s1 = 1
slli s1, s1, 2        # s1 = 1 << 2 = 4
beq  s0, s1, target   # branch is taken
addi s1, s1, 1        # not executed
sub  s1, s1, s0        # not executed
```

```
target:                # label
add  s1, s1, s0        # s1 = 4 + 4 = 8
```

Labels indicate instruction location (or data location). They can't be reserved words and must be followed by a colon (:)

RISC-V assembly

```
j      target          # jump to target
srai   s1, s1, 2        # not executed
addi   s1, s1, 1        # not executed
sub     s1, s1, s0      # not executed
```

```
target:
add     s1, s1, s0      # s1 = 1 + 4 = 5
```

Conditional Statements and Loops

If statement:

C:

```
if (i == j) {  
    f = g + h;  
}  
f = f - i;
```

RISC-V assembly:

```
# s0 = f, s1 = g, s2 = h, s3 = i, s4 = j  
bne s3, s4, L1  
add s0, s1, s2
```

```
L1:  
    sub s0, s0, s3
```

If else

C:

```
if (i == j) {  
    f = g + h;  
} else {  
    f = f - i;  
}
```

Assembly:

```
# s0 = f, s1 = g, s2 = h, s3 = i, s4 = j  
bne s3, s4, L1  
add s0, s1, s2  
j done
```

```
L1:  
    sub s0, s0, s3  
done:
```

While loop

C:

```
int pow = 1;  
int x = 0;  
  
while (pow != 128) {  
    pow = pow * 2;  
    x = x + 1;  
}
```

Assembly:

```
#s0 = pow, s1 = x
```

```
addi s0, zero, 1  
add s1, zero, zero  
addi t0, zero, 128
```

```
while:  
    beq s0, t0, done
```

```

    slli s0, s0, 1
    addi s1, s1, 1
    j while
done:

```

For loop for (initialization; condition; loop operation) { statement }

C:

```

int sum = 0;
int i;

for (i = 0; i != 10; i++) {
    sum += i;
}

```

Assembly:

```

# s0 = i, s1 = sum
    addi s1, zero, 0
    add s0, zero, zero
    addi t0, zero, 10
for:
    beq s0, t0, done
    add s1, s1, s0
    addi s0, s0, 1
    j for
done:

```

Less Than

C Code

```

// add the powers of 2 from 1
// to 100
int sum = 0;
int i;

for (i=1; i < 101; i = i*2) {
    sum = sum + i;
}

```

RISC-V assembly code

```

# s0 = i, s1 = sum
    addi s1, zero, 0
    addi s0, zero, 1
    addi t0, zero, 101
loop:
    slt t2, s0, t0
    beq t2, zero, done
    add s1, s1, s0
    slli s0, s0, 1
    j loop
done:

```

slt: set if less than instruction
 slt t2, s0, t0 # if s0 < t0, t2 =
 1

otherwise t2 = 0